MAPLE* TUTORIAL FOR MATH 302

Rakesh†

January 3, 2007

Contents

*Version 10.00

†Please send suggestions and corrections to rakesh@math.udel.edu

1 COMMANDS FOR MATH 242 MATERIAL

The commands given below have been tested only for the *Worksheet Mode* of Maple.

1.1 Differentiation, Integration, Editing

Start Maple. To differentiate $\sin(x^2)$ type

```
diff( sin(x^2), x);      hit Return
```

The ←, →, ↑, ↓ keys may be used to navigate in the worksheet and the *Backspace*, *Delete* keys may be used to correct typing errors. Don't forget the semicolon - every command ends with a semicolon. To compute definite and indefinite integrals use

```
int( sin(x), x= 0..Pi );       and hit Return
int( a*x^2, x );               and hit Return
```

Note that `Pi` (upper case P) is the symbol used for π in Maple. The first command computes the definite integral and the second command computes the indefinite integral. Notice, in the second command, ax^2 could have been integrated with respect to x or a.

One convenient editing trick is the following. Suppose we wish to differentiate $x\sin(x^3)$. We have already differentiated $sin(x^2)$ - so using the ← ↑ → ↓, keys on the right hand side of the keyboard, move the cursor to that line. Then using the *Backspace*, *Delete*, and arrow keys modify that line to read

```
diff( x*sin(x^3), x);         and hit Return
```

We now continue with other Maple commands - move the cursor to the last line using the ←, ↑, →, ↓. To compute higher order derivatives note the example below

```
f := sqrt(1+x) + 3*exp(x) - ln(1+x^2);
diff(f, x$3);
```

Here we first define an expression f and then we compute its third derivative (note the `x$3`). Also note the use of the square root function, the exponential function, and the natural log. **DO NOT** use `e^x` for the exponential function.

1.2 Destroying or Starting a Worksheet

To start a new worksheet, because you want to work on a new problem, *left* click the *File* menu and in the new menu which pops up *left* click the *New* line and choose *Worksheet Mode*. You will get a new worksheet called `Untitled 2`. To destroy the old worksheet called `Untitled 1`, click the ⊗ button associated with `Untitled 1` in the top left corner. Reply *No* to the question about saving the old worksheet.

1.3 Manipulating Windows

Open a Maple worksheet if you don't have one. To plot the graph of the function x^2 over the range $-1 \leq x \leq 1$, type

```
plot( x^2, x=-1..1 );
```

This will result in the plot of $y = x^2$ over $-1 \leq x \leq 1$.

The plot may be manipulated - click on the plot and a box appears around the plot; move the mouse arrow to the bottom right corner of the plot box and the arrow will turn into another shape; holding down the *left* mouse button drag that corner till you have resized the plot to your liking. From the *Plot* menu at the top, choosing the *Scaling Constrained* option will result in a plot in which the scale used in the x and the y direction are the same - the default is unconstrained scaling where the scales chosen are those which fill the box with the plot.

The menu appearing on the main Maple window depends on whether the plot box is active or not. Click on the `plot` command line and notice that the headings in the main Maple window change. Click on the plot again and see the change.

1.4 Getting Help in Maple

Maple has an excellent help facility. Suppose you wish to find out the Maple command to simplify an expression. Move to the the last Maple command line and type

```
?simplify
```

and a separate window pops up giving information about the `simplify` command. The most useful part of this *help* window are the examples at the bottom - scroll down to see them. Also useful is the line at the end which says *See Also* - this mentions other commands which may be appropriate. You can get information on any of the commands listed there by clicking on that word; for example, click on `collect` and analyze the examples at the bottom of the help window - seems like a useful command.

When you are done with the Help windows you can either destroy it or iconify it for later use. Destroy the Help window now. Help is also available from the menu at the top of the main Maple window - you can explore this further later.

1.5 Quitting Maple

To quit Maple, from the *File* menu choose `Exit`; answer `No` to prompts for saving this work (later we will see how to save your work). This will get you out of Maple.

1.6 Substituting values into expressions, Numerical value of an expression

Start Maple if you do not have a Maple window. To substitute a value for a variable in an expression use the command `eval`. To obtain a numerical value of an expression use `evalf`. We show their use by an example. We compute the third order x derivative of $x^2 \sin(kx^3)$ at $x = 1, k = 3$, accurate to 7 digits. Try

```
fxxx := diff( (x^2)*sin(k*x^3), x$3 ); computes 3rd order derivative
myval := eval( fxxx, {x = 1, k = 3} ); evaluate fxxx at x=1, k=3
evalf( myval, 7 );  obtain numerical value of myval accurate to 7 digits
```

evalf is the Maple command to obtain a numerical value for an expression and the answer is calculated accurate to 7 digits. To obtain greater accuracy change 7 to a higher number - say 9; try it. Note the use of curly parenthesis `{` and `}` , in `eval`, to specify all the variables that are being replaced.

1.7 Time Saving Tricks

Start Maple if you don't have a Maple window.

- Use space between characters *judiciously*, not too much and not too little, to make error detection easier. The following three lines do exactly the same thing (substitute values for variables in an expression) but the first is preferred over the other two. In the first line, note how the command, the expression and the variables are judiciously spaced so that the different units stand out.

  ```
  eval( sin(x+y), {x=2, y=3} );        Good
  eval(sin(x+y),{x=2,y=3});            correct but terrible
  eval( sin(x+ y),  {x =2,y =3} );     correct but room for improvement
  ```

- If the output of a certain calculation is to be used later, it is best to give it a name and choose a meaningful name. For example,

  ```
  myder := diff( x^2, x );
  int( myder, x );
  ```

- Occasionally Maple responds to a command with `"syntax error"`. e.g. try

  ```
  diff( x^3 + 3x^2 - x, x );
  ```

Maple responds with the error message. Also note that it points to the probable source of the error. Can you determine the error? The error is that `3x^2` should have been entered as `3*x^2`. Now instead of retyping the line it is quicker to correct the old line by using the arrow keys; then hit *Return*. Remember *modify only the red colored lines and never retype a line* - it is usually faster to correct the incorrect line.

- To use part of an old command one can use the *copy* and *paste* technique. Type

  ```
  diff( sqrt(1+x) + x*exp(x) + ln(1+x), x );                 Hit Return
  ```

  Now suppose we want to integrate the same function $\sqrt{1+x} + xe^x + \ln(1+x)$ from 2 to 5. Instead of retyping this expression we copy it. Highlight the expression to be copied (in the red colored part); then from *Edit* choose *Copy* (not *Cut*). Next, using the arrow keys, move the cursor to a new line, and try

  ```
  int( from Edit menu choose Paste, x = 2..5 ) ;
  ```

  You should obtain

  ```
  int( sqrt(1+x) + x*exp(x) + ln(1+x) , x = 2..5 ) ;           Hit Return
  ```

- In Maple, several commands may be executed together. This is particularly convenient if the desired output is the result of several commands and one is not interested in the intermediate results. For example, we will compute the value of

  $$\frac{d}{dx}\left(x^3 \sin(kx)\right) \quad + \quad \int_1^3 t\; cos(t+k)\; dt$$

  when $x = 2,\ k = 4$, accurate to 9 digits.

  ```
  a := diff( sin(k*x)*x^3 , x );   while pressing Shift hit Return
  b := int( t*cos(t+k), t=1..3 );  while pressing Shift hit Return
  c := eval( a+b, {x=2, k=4} );    while pressing Shift hit Return
  evalf(c, 9);                     Hit Return
  ```

  All the commands are executed together and the last result is the desired answer - make sure you understand why the last result is what we wanted. The problem was broken into several parts then combined to get the final result.

- In the previous problem, we were not really interested in the intermediate values a, b, c. In Maple one can suppress the output of a command by replacing the semicolon by a colon. In the previous commands replace the semicolon ; by a colon : for the *first three commands only* - these are the intermediate commands whose output does not interest us. Now hit return and note the output. Compare this with the earlier situation by putting back the semicolons.

Now is a good time to do the exercises below. They are quite straightforward and will help you assimilate the commands and ideas we have discussed so far.

Exercise 1.1 *Compute the fourth derivate of* $\cos(x+1)+\ln(x-1)+xe^{x^2+1}$ *with respect to* x*. Also compute the integral of the original function with respect to* x *over the interval* $[2,4]$ *- do not retype the expression to be integrated.*

Exercise 1.2 *What is the value of* $e^{u+v^2}+u^2\sin(uv)$ *when* $u=-3,\ v=2$*? Obtain this value accurate to 12 digits. What is the value of* $\sin 3$ *accurate to 9 digits? All trigonometric angles in Maple are in radians (and not degrees) by default.*

Exercise 1.3 *Using Maple determine the sine of 20 degrees, accurate to 10 digits. Hint: convert degrees to radians and then apply the sine function. Answer:* 0.3420201433 .

Exercise 1.4 *Determine the value of the following expression when* $p=3,\ q=1$*, accurate to 10 digits.*

$$\int_q^p u^2\ \sin(p\ u)\ du\ -\ \frac{d}{dq}\left(p\ q^3\sin(p\ q)+(p\ q+1)\ e^q\right)$$

Answer: -8.73549981 .

1.8 Interrupting Maple

If Maple gets stuck in a calculation to stop the calculation left click the *STOP* button. e.g. type

```
sum( sin(n^2), n=1..10^7 );   Hit Return
```

This attempts to find the sum of $\sin(n^2)$ as n ranges from 1 to 10^7. This will take too long so to stop the calculation left click the *red hand stop* button at the top of the window.

1.9 Some Common Errors

- Do not type xy when you really want $x*y$, or type $\cos x$ when you want $cos(x)$.
- A common mistake is the inappropriate use of the `solve` command. The `solve` command is useful only for solving equations (explained later) - it is useless/incorrect to use it for simplifying expressions (use `simplify`) or obtaining numerical values for expressions (use `evalf`).

- A common error is to use a symbol as a variable even though the symbol was assigned a value earlier. For example, suppose at some stage you used $x := 4$; and you forgot about it. Some commands later you differentiate $\sin x$ expecting $\cos x$ but instead you get an error message because x has the value 4. One clears the value of x is shown below. Try

```
x := 4;
diff( sin(x), x );                              Error
x := 'x';  use quote key right of semicolon key, this clears value of x
diff( sin(x), x );           now it works as you want it to
```

- Since the commands in a Maple worksheet are often modified and then rerun, it is a good idea to start every worksheet with the `restart` command. Execute the following commands -

```
diff( sin(x), x);
diff( cos(y), y);
x := 4;
```

  Now reexecute each of the commands by just hitting *Enter*. You will notice a problem with the first command. That is so because Maple remembered that $x = 4$ from the third command so in the rerun the first command does not make sense. So try the following - move the cursor to the first line, then from *Insert* choose *Execution Group* choose *Before Cursor* and you should have a new execution group as your first line. In that line type `restart;` and then reexecute the next set of commands and there should be no problem.

1.10 Plotting Planar Curves - I

Planar curves may be given explicitly as in $y = x^2$ (a parabola), implicitly as in $x^2 + y^2 = 1$ (a circle), or in parametric form as in $x = t + \sin t,\ y = 1 - \cos t,\ -6 \leq t \leq 6$ (a cycloid). Curves may also be given in polar form as in the three leaved rose $r = \sin(3\theta)$.

To draw the graph of the explicitly given curve $y = x^2 + 9\sin x$ over the interval $[-3, 3]$ use

```
plot( x^2 + 9*sin(x), x=-3..3 );
```

To draw the graph of two (or more) explicitly given curves $y = x - x^2$ and $y = 1 - 2x$ over the interval [0,1], in the same window, with $y = x - x^2$ colored red and $y = 1 - 2x$ colored blue, use

```
plot( [x-x^2, 1-2*x], x=0..1, color=[red,blue] );
```

To determine the point of intersection of the two curves in $0 < x < 1$, *left click* the point of intersection to obtain its approximate location as $(0.38, 0.23)$ - this is displayed in the box in the left corner of the main window.

To draw curves given implicitly (i.e. they are *not* given in the form $y = f(x)$, but in the form $g(x, y) = c$) use the *implicitplot* command. However, this command is in the *plots* package; so to draw the ellipse $x^2 + 4y^2 = 4$ use

```
with(plots);        Load the plots package - only once in a worksheet
implicitplot( x^2 + 4*y^2 = 4, x=-3..3, y=-3..3 );
```

Note that for `implicitplot`, both the x and y ranges have to be given. `implicitplot` plots the part of the curve which lies in the rectangle determined by the x and the y ranges. Compare this with plotting using the plot command (for example to plot $y = x^2$) where only the x range is required. Examine what happens if you increase or decrease the ranges for x and y - try several changes. For `implicitplot`, it is advisable to start with a large range and then decrease the range until the graph is satisfactory.

To draw more than one implicitly given curve, for example, the ellipse $x^2 + 4y^2 = 4$ and the circle $x^2 + y^2 = 2$, with ellipse blue and circle colored green, use

```
implicitplot( [x^2 + 4*y^2 = 4, x^2 + y^2 = 2],        Shift Return
                x=-3..3, y=-3..3, color=[blue, green] );
```

Again the approximate locations of the points of intersection may be read with the help of the mouse. Some remarks are in order here.

- Curves which can be plotted using `plot` may also be plotted using `implicitplot` but not vice versa. Wherever possible use `plot` instead of `implicitplot` because `plot` is faster and gives more accurate graphs.
- Notice the use of the curly brackets `[ ]` when plotting two or more curves - all the equations/functions being plotted must be enclosed in these square brackets. These square brackets are not necessary (but may be used) when plotting a single curve.

To superimpose two plots (obtained from different commands) use the `display` command in the *plots* package. For example, to display the graphs of $y = x^2$, $y = 2x$, and the circle $x^2 + y^2 = 1$ in the same picture, use

```
with(plots):
pic1 := plot( [2*x, x^2], x=-2..2, color=[red, blue] );
pic2 := implicitplot( x^2 + y^2 = 1, x=-2..2, y=-2..2, color=green);
display( {pic1, pic2} );
```

The second and third lines define two graphs named *pic1* and *pic2*. The last line displays the two pictures - superimposed.

1.11 Exact Solution Of Systems Of Equations

To obtain solutions of one equation in one variable or to find the solutions of a system of equations, the `solve` command is useful. To find all solutions of the quadratic equation $x^2 - 3x + 2 = 0$, use

```
solve( x^2 - 3*x + 2 = 0, x );
```

In the command `solve`, the first parameter is the equation to be solved and the second parameter is the variable to be solved for.

To solve the system of equations $x + y = 1$, $x^2 - y = 1$, for the variables x, y, use

```
solve( {x + y = 1, x^2 - y = 1}, {x,y} );
```

and we obtain the two sets of solutions.

Now consider the following example - find all solutions of $3x - 2y = 5, \;\; x^2 - y^3 = 2$. As before, we use

```
mysol := solve( {3*x - 2*y = 5, x^2 - y^3 = 2}, {x,y} );
```

where `mysol` is the name given to the list of solutions generated. The solutions are $\{x = 1,\ y = -1\}$ and $x = Z,\ y = 3Z/2 - 5/2$ where Z is one of the roots of the quadratic equation $9Z^2 - 13Z - 7 = 0$. Now the quadratic equation has two solutions - so there are two choices for Z. To obtain both these solutions coming from the Z, we first isolate the part containing Z. Try

```
mysol[1];          first part of mysol
mysol[2];          second part of mysol
```

So `mysol[2]` contains the terms dependent on Z. To get all the solutions (that is expanding the `Rootof` part) apply the `allvalues` command to `mysol[2]` .

```
othersolns := allvalues( mysol[2] );
```

So we have the other two solutions of the system of equations - giving a total of three solutions. To obtain the numerical values of the other two solutions accurate to 5 digits, we can apply `evalf`

```
evalf( othersolns, 5 );
```

Often the solutions of a system of equations are to be substituted in another expression. This can be done very efficiently as shown below. Suppose we wish to find the value of $\ln(x^2 + y^2)$ at the solution of above system which lies in the first quadrant (that is $x > 0$ and $y > 0$). Of the three solutions obtained above, it is the $x = 2.9080,\ y = 1.8621$ solution which lies in the first quadrant. So one could use

```
eval( ln(x^2+y^2), {x=2.9080, y=1.8621} );
```

However, this is not quite correct and also inefficient. Firstly, $x = 2.9080,\ y = 1.8621$ is only an approximation to the solution, the correct solution being $y = 13/18 + \sqrt{421}/18,\ x = \cdots$, so we should be substituting this into $\ln(x^2 + y^2)$ and not its numerical approximation. The second more important issue is that one should not have to retype results which Maple has generated - instead the result should be given a name and the name should be used.

In our case, the solution of interest is `othersolns[1]`, try

```
othersolns[1];
```

Note that the solution is already in the form $x = \cdots,\ y = \cdots$, exactly the form needed for using the command `eval`, so a more efficient way of getting the value of $\ln(x^2 + y^2)$ at this solution would be

```
myval := eval( ln(x^2+y^2), othersolns[1] );
evalf( myval, 10 );                         numerical value
```

Please try the exercise below before moving to the next section. The ideas in this section are very useful.

Exercise 1.5 *Find the numerical value of* $\sin(x^2 + x)$ *where* x *is the positive root of the equation* $x^3 + 3x^2 = 2$. *Answer :* 0.9544912329.

Exercise 1.6 *Evaluate* $x^3 + e^{xy}$ *at the solution of* $y = x + 1,\ x^2 + y^2 = 4$ *which lies in the first quadrant. Ans:* 5.038878215 .

1.12 Approximations Of Solutions Of Equations

To find all solutions of the equation $7\cos x + x + x^2 = 15$ try

```
solve( 7*cos(x) + x + x^2 = 15, x );
```

Maple tries to find the exact solutions and fails. So we attempt to obtain the approximations to these solutions using the command `fsolve`.

First we must find out how many solutions this equation has. For large $|x|$, $7\cos x+x+x^2$ is very large and hence will not equal 15. Hence the equation will not have a solution x for which $|x|$ is large; so the solutions of $7\cos x + x + x^2 = 15$ are in a finite interval. The solutions of $7\cos x + x + x^2 = 15$ are solutions of $7\cos x + x + x^2 - 15 = 0$, that is the values of x where the graph of $7\cos x + x + x^2 - 15$ cuts the x axis (WHY?). So we plot the graph of $7\cos x + x + x^2 - 15$ over a reasonably large interval.

```
plot( 7*cos(x) + x + x^2 - 15, x=-10..10 );
```

From the graph, using the mouse, we observe that the graph cuts the x axis at two points - hence the equation has two solutions, one between -5 and -4, and the other between 3 and 4. To find these solutions we use the `fsolve` command. So we can obtain the approximate value of the two solutions using

```
fsolve( 7*cos(x) + x + x^2 = 15, {x}, x=-5..-4 );
fsolve( 7*cos(x) + x + x^2 = 15, {x}, x=3..4 );
```

The `fsolve` command needs the equations and the variables just as `solve` but `fsolve` also needs an interval which encloses a solution. When using `fsolve` it is important to choose an interval which contains only the solution one wants - the narrower the interval the better. This is usually done by first plotting the appropriate graphs - so *always draw the appropriate graphs before using* `fsolve`.

Finally, to complete the solution of the problem we must show somehow (using pencil and paper) that $7\cos x + x + x^2 = 15$ has no solutions outside $[-10, 10]$. That takes some effort and we will skip that.

Exercise 1.7 *Find all the solutions of $x^2 + 2x + x\sin x = 4$. Check your answer by substituting these values of x into the equation.*

We now study how to find solutions of systems of equations for which `solve` fails to work. Consider the system of equations $x^2 + y^2 = 4$ and $\sin(x + y) + \cos x = 1$. If we try

```
solve( {x^2 + y^2 = 4, sin(x+y) + cos(x) = 1}, {x,y} );
```

then Maple is unable to find the exact solutions. We plot the two curves $x^2 + y^2 = 4$ and $\sin(x+y)+\cos(x) = 1$ and the solutions of this system of equations are the points which lie on both the curves.

```
with(plots);    unnecessary if already done once this session
implicitplot( [x^2 + y^2 = 4, sin(x+y) + cos(x) = 1], x=-3..3, y=-3..3 );
```

Using the mouse, we note that the system has two solutions (in the rectangle $[-3,3]\times[-3,3]$) - one near $(-1,1.6)$ and the other near $(1,1.7)$. The system will not have any solutions outside the rectangle $[-3,3] \times [-3,3]$ - WHY? To find these solutions more accurately use

```
fsolve( {x^2 + y^2 = 4, sin(x+y) + cos(x) = 1}, {x,y}, x=-2..0, y=0..2 );
fsolve( {x^2 + y^2 = 4, sin(x+y) + cos(x) = 1}, {x,y}, x=0..2, y=0..2 );
```

Again, `fsolve` needed the system of equations, the variables to be solved for, and the range over which the search for the solution is to be conducted. The ranges for x and y were chosen carefully - using the graph, they were chosen to include only one solution at a time. So again, *plot the appropriate curves before using the* `fsolve` *command (to help determine the ranges to be used in* `fsolve`*).*

1.12.1 Using the *avoid* option with *fsolve*

The graphical technique, of obtaining the rectangle containing the solution we want, fails in three and higher dimensions (i.e. solving k equations in k unknowns with $k \geq 3$) because the graphs are hard to visualize in three dimensions and non-existent visually in higher dimensions; however, `fsolve` will still work.

```
sol1 := fsolve( {x^2 + y^2 + z^2 = 4, x+y+z=0, x*sin(y*z)=-1},  Shift Return
        {x,y,z}, x=-2..2, y=-2..2, z=-2..2 );
```

gives one solution of the system of equations.

Is it the only solution? How do we find others if there are any? Maple's `solve` command has an option where it finds solutions which avoid certain specified solutions. Try

```
sol2 := fsolve( {x^2 + y^2 + z^2 = 4, x+y+z=0, x*sin(y*z)=-1},  Shift Return
        {x,y,z}, avoid = sol1 );
```

which gives a solution different from `sol1`. Are there more solutions? We try

```
sol3 := fsolve( {x^2 + y^2 + z^2 = 4, x+y+z=0, x*sin(y*z)=-1},  Shift Return
        {x,y,z}, avoid = {sol1,sol2} );
```

to find solutions which avoid `sol1, sol2` and it seems that there are none. Use `?fsolve/details` to find about other options for `fsolve`.

Exercise 1.8 *Find all solutions of* $x^2 + y^2 = 9, \quad x^3 + y^3 - \sin(xy) = 7$. *Why are you sure that there are no more? Answer: There are two solutions.*

1.13 Commands Longer Than A Single Line

As we saw in the previous example, some commands need more space than a single line. To split up a command over several lines do as below

```
solve( {x+y+z=2, x+2*y+3*z=7, x-2*z=10},  holding down Shift press Return
       {x,y,z} );
```

1.14 Procedures

So far we have used expressions instead of functions. For example, if we want to work with the function $x^3 + \sin x$ then we have chosen to use

```
p := x^2 + sin(x) ;
```

However, the `p` defined above is an expression and not a function. If we want the value of that expression at some point then we use the `eval` command. A better way may be to define a Maple function or procedure as

```
f := proc(x)            Shift Return
     x^2 + sin(x)       Shift Return
     end ;
```

Every Maple procedure has the form

```
name of procedure := proc( variables separated by commas )
                           value of the function
                     end ;
```

Given next is an example of a procedure dependent on two variables

```
g := proc(x,a)                     Shift Return
     x^3 + a^3 + x*sin(a) ;        Shift Return
     end;                          Return
```

Examine the following use of the procedures f and g carefully and make sure that the output of each command is what you expect it to be.

```
f(2);               exact value of f at x=2
f(2.0) ;            numerical value of f(2)
fp := D(f);         fp is the derivative of f - it is also a procedure.
D(f)(3);            exact value of derivative of f at x=3
D(f)(3.0);          numerical value of derivative of f at x=3
plot(f, -2..2);     graph of f over the interval [-2,2]
g(2,1);             exact value of g at x=2, a=1
g(2.0, 1.0);        numerical value of g at x=2, a=1
D[1](g);   D[2](g);
```

Procedures are very useful in implementing algorithms. We will see the use of procedures when implementing Newton's method and Simpson's rule.

1.15 Finite sequences and series, Taylor Polynomials

To find the sum of the series

$$\sum_{n=5}^{69} \frac{\sin n}{1+3n^2}$$

accurate to seven digits, use

```
a := Sum( sin(n)/(1+3*n^2), n=5..69 );
evalf(a,7);
```

The first statement assigns a the sum of the series; the second obtains a 7 digit numerical value for a. The upper case `S` in the `Sum` command tells Maple to return only the symbolic sum. Repeat the above commands after replacing `Sum` by `sum` - see what happens.

A command useful for generating sequences of numbers, expressions, etc., arising from one 'formula' is `seq`. Observe the result of executing the following statements

```
seq( i^2, i=3..7 );
seq( i^2, i={3,4,10,20} );
seq( sin(k*x), k=1..4 );
```

For example to plot the graphs of $\sin x,\ \sin(2x), \cdots, \sin(5x)$ over the interval $[-\pi, \pi]$ one could use

```
funs := seq( sin(k*x), k=1..5 );
plot( [funs], x=-Pi..Pi );
```

Of course, it would be nicer if we knew which graph corresponded to which function. This may be done as follows - click on the plot, and then from the `Legend` menu choose `Show Legend` and the graphs are labeled.

The Maple command to find the Taylor expansion of a function is `mtaylor`. To find the Taylor expansion of $\ln x$ around $x = 1$ of order 6, use

```
mtaylor( ln(x), x=1, 7 );        use 7 to get expansion up to order 6
```

1.16 Manipulating Expressions, Extracting Parts of Expressions

Maple has many commands to manipulate algebraic expressions. A few of them are `collect`, `expand`, `simplify`, `combine` - use the `?` to find out more about them. To extract parts of an expression two useful commands are `op` and `nops`.

Consider the expression $(x + py)^3 + (2x - y)(3x^2 + y^2 - px)$. If we want the expansion of this in powers of x and y then we would use

```
restart:
f := (x+p*y)^3 + (2*x-y)*(3*x^2 + y^2 - p*x);
g := simplify(f);   if simplify had not worked we would try expand
h1 := collect(g, x);
h2 := collect(g, {x,y});
h3 := collect(g, {x,y}, distributed);
```

Note the three different ways of collecting the terms in `g` (or `f`); `h1` organizes the terms of g by powers of x alone ignoring the powers of the other variables; `h2` organizes the terms of g first by powers of x and then the coefficients of each power of x are organized by powers of y; where as in `h3`, g is organized in powers of x and y as a polynomial (or more correctly a multinomial). Each of these representations is useful depending on what we use we have in mind. Expressions may be organized not just in powers of x but in powers of any expression. Try

```
k := (x+sin(x)) * (3*x + sin(x)) + (2*sin(x) + x)^2 ;
collect( expand(k), sin(x) );
```

Now we reexamine the expression `h3` obtained earlier. We will extract various pieces of `h3`. Suppose we want to extract an expression which consists of the first three terms in `h3` and another expression which consists of the rest of the terms in `h3`. We will be using two commands - `nops(a)` which counts the number of terms/operands in the expression `a`, and `op(k,a)` which returns the kth term/operand of the expression `a`. So

```
h3;   just to see the expression
nops(h3);   counts the number of terms in h3 (not needed)
op(1,h3);   first term in h3 (not needed)
op(2,h3);   second term in h3 (not needed)
a := op(1,h3) + op(2,h3) + op(3,h3);    the first three terms of h3
b := h3 - a;    the rest of the terms of h3
```

To extract all the coefficients of the various powers of x, y in `h3` use

```
coeffs(h3, {x,y});
```

It is a little trickier to get just the coefficient of a single term such as x^2y. We use Maple's `coeff` command which works only on expressions expanded in terms of one variable.

```
h3; just to see the expression, not necessary
t1 := collect(h3, {x,y});  expand h3 in powers of x (then coeff in powers of y)
t2 := coeff(t1, x^2); extract terms which have an x^2
t3 := coeff(t2, y); extract term with y  - hence the x^2 y term is extracted
```

Exercise 1.9 *Consider the expression*

$$(ax^2 + bx\sin y + c\sin y)^2 \ + \ (a\sin^2 y + bx)^3$$

i) Write the above as a polynomial in x and extract the coefficient of x^2. Ans : $b^2(3a + 1)\sin^2 y + 2ac\sin y$
ii) Write the above as a multinomial in the variables x and $\sin y$ and extract the coefficient of $x\sin^2 y$. Ans : $2bc$.

1.17 Plotting Planar Curves - II

We now learn the commands for plotting planar curves given in parametric form or in polar coordinates. To draw curves given in parametric form one still uses the plot command. For example, to plot the *cycloid* $x = t + \sin t, \quad y = 1 - \cos t$ use

```
plot( [t+sin(t), 1-cos(t), t=-6..6] );
```

where the range $t = -6..6$ was chosen somewhat arbitrarily. What happens if you change the range for t - try it. Also, the location of the range of `t=-6..6` on the command line is very important. If the part `t=-6..6` is moved out of the `[ ]` then the graphs obtained are different. Try

```
plot( [t+sin(t), 1-cos(t)], t=-6..6 );
```

You will obtain the graph of two curves $y = t + \sin t$ and $y = 1 - \cos t$ as t varies over $[-6, 6]$ - see the section 1.10 where this was discussed.

To draw the two (or more) parametric curves, the cycloid $x = t + \sin t, \quad y = 1 - \cos t$, and the circle $x = 2\cos t, \quad y = 2\sin t$, use

```
plot( [ [t+sin(t), 1-cos(t), t=-6..6], [2*cos(t), 2*sin(t), t=0..7] ], Shift Return
        color=[red,green] );
```

The circle does not really look circular but that may be corrected by choosing the *Constrained* option for the plot. The ranges $t = -6..6$ and $t = 0..7$ were chosen somewhat arbitrarily. What is the effect of decreasing or increasing these ranges? Try it and *observe* what happens and why.

To plot curves given in polar coordinates use the command `polarplot` in the *plots* package. To draw the circle $r = \cos\theta$ (with $0 \le \theta \le \pi$) and the cardiod $r = 1 + \cos\theta$ (with $0 \le \theta \le 2\pi$) in the same picture, use

```
with(plots);
polarplot( [ [cos(t), t, t=0..Pi], [1+cos(t), t, t=0..2*Pi] ] );
```

Here we have used the symbol t instead of θ; choose the *Constrained* option for the plot.

To superimpose two plots (obtained from different commands) use the `display` command in the *plots* package. For example, to display the graphs of $y = x/2$ and the cardiod $r = 1 + 2\cos\theta$ in the same picture, use

```
with(plots):
pic1 := plot( x/2, x=-1..4, color=red );
pic2 := polarplot( [1+2*cos(t), t, t=0..2*Pi], color=green );
display( {pic1,pic2} );
```

The second and third lines define two pictures *pic1* and *pic2* and the last line displays the two pictures - superimposed.

There are some interesting plotting commands in the *plottools* package. Use `?plottools` to find out more about it.

Exercise 1.10 *Generate a plot showing the three leaved rose* $r = \sin 3\theta$ *and the circle* $r = 1$ *- coloring the rose red and the circle blue. Do the same showing only the right hand side of the rose and the full circle. Rewrite the two curves in parametric form* $x = ..$, $y = ..$ *and plot them using the* `plot` *command (instead of the* `polarplot` *command).*

1.18 Summary

We have used the following commands in the tutorial so far.

`diff, int` — differentiate, integrate
`evalf` — getting a numerical approximation of a number
`eval` — evaluating an expression when variables are assigned values
`solve, allvalues, fsolve` — solving systems of equations
`plot, implicitplot, polarplot, display` — plotting curves in 2D
`simplify, collect, expand, op, nops` — manipulating and extracting parts of expressions
`sum` — summing the terms of a finite series
`seq` — generates the terms of a sequence
`mtaylor` — Taylor polynomial of a function
`with` — loading a package

We have seen three kinds of parentheses used in Maple - ( ), { }, [ ]; each plays a different role.

- ( ) is used in expressions as in $(x - y) * (x + 7 * y)$, or surrounds the argument of a standard function as in $sin(x + y)$ or surrounds the arguments of a Maple function as in

```
solve ( ...usual stuff here... );
```

- { } is used in lists of expressions, equations, or variables where the order of the listing is unimportant, as in

```
solve( {x + y = 1, 2*x + 7*y =0}, {x,y} );
eval( x^2 + z^3, {x=2, z=4} );
```

- [ ] is used in *ordered* lists as in

```
plot( [t+sin(t), 1-cos(t), t=-6..6] );
```

when plotting the parametric curve $x = t + \sin t,\ y = 1 - \cos t$. Note the order of the elements in this list is important - the first element representing the x variable, the second the y variable etc. Another situation where order is important and where `[ ]` is used is in

```
plot( [x, x^2], x=-1..1, color=[red, blue] );
```

Here the graphs of $y = x$ and $y = x^2$ are plotted and the order is important because $y = x$ is colored red and $y = x^2$ is colored blue - to help use recognize which graph is which.

`[ ]` is also used to refer to one of the solutions, of a system of equations, obtained by using the *solve* command; see subsection 1.11 ; [ ] is also used in vectors and matrices.

1.19 Problems

1. Graphically determine the number of solutions of the equation
$$x^2 - x\sin(3x) - x\ln(x+2) = 1$$
in the interval [0,3] and the approximate values of these solutions. Then find the solution closest to $x = 2$ accurate to 6 digits. *Ans :* $x = 1.3, 2.2, 2.8;\ x = 2.199377$.

2. Find the following sum accurate to 8 digits
$$\frac{1}{2} + \frac{1}{3} + \frac{1}{4} + \cdots + \frac{1}{100}$$
Ans: 4.1873775.

3. Find all solutions of the system of equations
$$\begin{aligned} x^2 + y^2 &= y \\ 2x + 2y^2 + 2\cos(xy) &= 3 \end{aligned}$$
which lie in the region $-1 \le x \le 1,\ -2 \le y \le 2$. Your answer should be accurate to at least four digits. *Ans:* $(x = -0.2895,\ y = 0.9076)$ *and* $(x = 0.43877, y = 0.26026)$.

4. Find the largest and the smallest value of the function $x^2 - x\sin(3x) - x\ln(x+2)$ over the interval $[0, 3]$; at which points are these values attained. Your answer should be accurate to four digits. *Ans:* $max = 2.93533$ *at* $x = 3$; $min = -0.81567$ *at* $x = 0.6639$.

5. Let $f = A\cos x + B\sin x + Px^2 + Qx + R$. Find the value of A, B, P, Q, R so that
$$\frac{d^2 f}{dx^2} + 3\frac{df}{dx} + 2f = 5\sin x + 2x^2 + 1$$
Ans : $a = -3/2,\ b = 1/2,\ p = 1,\ q = -3,\ r = 4$.

6. Find all solutions of the system of equations

$$\begin{aligned} x^2 + y^2 + z^2 &= 9 \\ x + y + z &= 2 \\ x + 2y + 3z &= 5 \end{aligned}$$

Then compute the value of $exp(xy+z)+z^2$ at the solution where z is the largest. *Ans: two solutions, largest z solution is* $(x = 1.25958, y = -1.51914, z = 2.25958)$, *value is* 6.5192.

2 Preparing a Report With Graphs, Comments, and Commands

When submitting a report of your Maple work you may wish to include comments and graphs between the commands. Below we give an example of preparation of a report which presents the solution of the problem

Find all solutions of the equation $x \sin x = 5$ *in the interval* $[0, 10]$.

- Open a new worksheet by choosing *New* and then *Worksheet Mode* from the File menu.
- We first provide a *heading* for our report. From the *Insert* menu choose *Section*. Next, from the *Insert* menu choose *Paragraph Before Cursor* to insert a paragraph before the cursor. Next move cursor to the top red arrow and delete that execution group by choosing *Delete Element* from the *Edit menu*. Then change the font size from 12 to 18, click the `B` to use boldface, and from the *Format* menu choose *Paragraph* and then *Center* to center your text. Now we may type our heading. Type

```
First Report For Math 242/243/302 (hit  Shift Return)
Your name (hit Shift Return)
The date (hit Shift Return)
```

- A report may contain the solution of one or more problems. So in most cases each problem must be solved in a separate section of the report. To start a section, from *Insert* choose *Section*. We give this section a heading - type

```
Finding all solutions of x sin(x)=5 in [0,10]
```

- From *Insert* choose *Execution Group After Cursor* and we get the prompt `[>`.
  Then from *Insert* choose *Paragraph Before* cursor to enter a comment about what you will do next. Type without hitting *Return*

```
We start by finding the approximate location of the solutions by
plotting the graph of x sin(x)-5
```

and then move the cursor to right of `>` to be ready to type the Maple commands. Now type

```
restart;                                   Hit Shift Return
plot( x*sin(x)-5, x=0..10 );
```

Notice the above is typed in red - the color for Maple commands. Hit *Return*. The places where the plot crosses the x-axis gives the approximate locations of the solutions of $x \sin x = 5$ in the interval $[0, 10]$. Reading off these points we find that one solution is between 6 and 8 and the other is between 8 and 10. We enter this as a comment as follows

- Move the cursor to the new line generated which contains a `[>`.
  From *Insert* choose *Paragraph Before* and type

```
We observe the solutions are between 6 and 8, and 8 and 10. So the
solutions are obtained by
```

Now move cursor to position after `>` .

- Now type the command to get the solutions

```
sol1 := fsolve( x*sin(x)=5, x=6..8 );      Hit Shift Return
sol2 := fsolve( x*sin(x)=5, x=8..10 );         Hit Return
```

and we obtain the two solutions. We state our conclusion

- The cursor should again be positioned after `>` in the new line. From the *Insert* menu choose *Paragraph Before* and then type the comment

```
The equation x sin(x) = 5 has two solutions in [0,10]. Accurate to
8 digits they are 7.0688914 and 8.8222213.
```

and our report is complete.

3 ODES AND MAPLE

Maple may be used to solve ODEs, plot their solutions, and draw their direction fields. This handout describes the Maple commands which help us do it. Additional help is available via Maple's `?` command - pay special attention to the examples when you use `?` .

3.1 Exact and Numerical Solutions of a Single ODE

3.1.1 Exact Solutions

Suppose we wish to solve the ODE $dy/dx + y/x = 1$. We first enter the differential equation and then use the `dsolve` command to solve the differential equation.

```
restart:
myode := diff( y(x),x ) + y(x)/x = 1;  enters the ode and names it myode
gensoln := dsolve( myode, y(x) );      general solution of the ode
```

Note that `gensoln` has one arbitrary constant because we did not prescribe any initial condition for $y(x)$. To find the solution of `myode` with the initial condition $y(1) = 3$ we can still use `dsolve`

```
newsoln := dsolve( {myode, y(1)=2}, y(x) );
```

Generally, `dsolve` attempts to write the solution in explicit form, that is as $y(x) = \cdots$ unless it is unable to do so - then it gives the solution implicitly. If the solution is given in explicit form as, say, $y(x) = \cdots$ then the RHS of this expression is what we want. This can be extracted by the `rhs` command of Maple. For example, try

```
anotherode := diff(y(t), t) + y(t) = sin(t);
soln := dsolve( {anotherode, y(0) = 3}, y(t) );
f := rhs(soln);   solution is assigned the name f for future use
```

If the explicit solution of the ODE is too cumbersome, one may prefer the more compact implicit solution form. Let us try this with the general solution of the ODE $(1 - y^2)y' = x^2$.

```
myode := (1-y(x)^2)*diff(y(x), x) = x^2 ;
soln := dsolve( myode, y(x) );    it is very messy
impsoln := dsolve( myode, y(x), 'implicit' );  implicit form of solution
```

Note the use of the word `'implicit'` as one of the options in the dsolve command; clearly the implicit form of the solution is much nicer in this example.

The `dsolve` command works in the same fashion for 2nd order (or higher) order ODEs. For second order ODEs prescribing initial conditions involves prescribing $y(a)$ as well as $y'(a)$ - pay particular attention to how the initial conditions involving derivatives is handled. To find the general solution and the solution of the ODE

$$y'' + 3y' + 2y = x^2$$

with the initial conditions $y(1) = 2$, $y'(1) = 3$, we use

```
myode := diff(y(x), x$2) + 3*diff(y(x), x) + 2*y(x) = x^2;
dsolve( myode, y(x) );   the general solution
dsolve( {myode, y(1)=2, D(y)(1)=3}, y(x) );   solution with y(1)=2, y'(1)=3
```

For third order ODEs the initial conditions would require prescribing $y(a), y'(a), y''(a)$. So to find the general solution and the solution of the ODE

$$y''' - y'' + y' - y = \sin x$$

with the initial conditions $y(0) = 1$, $y'(0) = 2$, $y''(0) = -1$, we use

```
newode := diff( y(x), x$3) - diff(y(x),x$2) + diff(y(x),x) - y(x) = sin(x) ;
dsolve( newode, y(x) );   general solution of newode
dsolve( {newode, y(0)=1, D(y)(0)=2, (D@@2)(y)(0)=-1}, y(x) );  solution of
newode with y(0)=1, y'(0)=2, y''(0)=-1
```

Note the use of $D(y)(1)$ and $(D@@2)(y)(0)$ to represent $y'(1)$ and $y''(0)$

3.1.2 Numerical/Approximate solutions

One cannot obtain the exact solution of most ODEs just as most functions cannot be integrated. However, one can obtain approximate (numerical) solutions to these ODEs. Consider the ODE

$$\frac{dx}{dt} = \frac{x+t}{1+x^2t^2};$$

here x is the dependent variable and t is the independent variable. If we try to find the general solution, or the exact solution with the initial condition $x(0) = 1$, using

```
newode := diff( x(t),t ) = ( x(t)+t )/( 1+ (x(t)*t)^2 );
dsolve( newode, x(t) );
dsolve( {newode, x(0)=1}, x(t) );
```

we fail because Maple is unable to find them for this ODE using any of the tricks it knows, or maybe the solution of this ODE can not be given in terms of the standard functions such as sin, cos etc. So we attempt to obtain the approximate solution. To find the approximate (i.e. numerical) solution we must prescribe an initial condition just as numerical integration requires numerical upper and lower limits of integration. So we now determine the numerical solution of `newode`, defined above, with the initial condition $x(0) = 1$.

```
mysoln := dsolve( {newode, x(0)=1}, x(t), 'numeric' );
```

note the use of the `'numeric'` option. When the `'numeric'` option is used in `dsolve` then the result is not an expression (which is what we have been getting so far) but a procedure named `mysoln` which will return values when a t is prescribed. So to find the value of the solution $x(t)$ when $t = 2$ or $t = 7.5$ we would use

```
mysoln(2);        approximate value of x(2)
mysoln(7.5);      approximate value of x(7.5)
```

Numerical solutions of 2nd and higher order ODEs may be constructed in a similar fashion. Occasionally, Maple has difficulty finding numerical solutions using the default methods - use Maple's help facility to get more information about the `numeric` option of `dsolve`, when you face such a problem.

3.1.3 Plotting solutions and direction fields

One can plot solutions of ODEs without calculating the exact solution because Maple can compute the numerical solution. Plotting is done with the command `DEplot` in the `DEtools` package. To plot the three different solutions of

$$\frac{dx}{dt} = \frac{t^2 - x^2}{x^2 + t^2}$$

corresponding to the three different initial conditions $x(1) = 2$, $x(1) = 0$, $x(1) = -1$, over the t range $0 \le t \le 10$, we use

```
with(DEtools);
myode := diff( x(t),t ) = (t^2 - x(t)^2)/(t^2 + x(t)^2) ;
DEplot( myode, x(t), t=0..10, { [x(1)=2], [x(1)=0], [x(1)=-1] } );
```

This generates a plot of the solutions (as well as the direction field for 1st order equations) corresponding to the ODE. If you feel that the arrows of the direction field clutter up the figure then they may be turned off by setting `arrows=NONE`, so modify the previous command to

```
DEplot( myode, x(t), t=0..10, { [x(1)=2], [x(1)=0], [x(1)=-1] }, arrows=NONE );
```

There are some other useful options in `DEplot`. The plots generated by `DEplot` are obtained by sampling the numerical solution at 20 points in the t interval we have chosen; in this example $0 \leq t \leq 10$ so the solution is sampled every $(10-0)/20 = 0.5$ units. To get more detail in the plot we may choose to sample more frequently, which is done by the option `stepsize=0.01` say. So modify the last command to

```
DEplot( myode, x(t), t=0..10, { [x(1)=2], [x(1)=0], [x(1)=-1] },
    stepsize= 0.01, arrows=NONE);
```

Try increasing the stepsize to 2.0 - what happens. Another useful option is `linecolor`. While one may determine which solution curve corresponds to which initial condition by examining the value of $x(t)$ when $t = 1$, it would be more convenient to provide different colors to the three solutions. This may be done as follows

```
DEplot( myode, x(t), t=0..10, { [x(1)=2], [x(1)=0], [x(1)=-1] },
   arrows=NONE, linecolor=[red,blue,green]);
```

So the trajectory corresponding to the first initial condition is colored red, the second one blue, and the third one green. To make this more obvious click the graph, and then from the `Legend` menu choose `Show Legend`. Other useful options for `DEplot` may be discovered by getting help on `DEplot`.

To plot only the direction fields of a first order ODE such as $y' = y^2 \sin x$ over the region $-5 \leq x \leq 5$, $-3 \leq y \leq 3$ use the `dfieldplot` command in the `DEtools` package.

```
myode := diff(y(x),x) = sin(x)*y(x)^2;
dfieldplot( myode, y(x), x=-5..5, y=-3..3, arrows=SLIM );
```

Exercise 3.1 *For the ode* $y' = \sin(x^2 + y^2)$

1. *Plot the direction field over the region* $-2 \leq x \leq 2$, $-2 \leq y \leq 2$.
2. *In the same plot, plot the solutions corresponding to the initial conditions* $y(0) = -1$, $y(0) = 0$, *and* $y(0) = 1$ *- also color them red, blue, and yellow respectively. Does the solution follow the direction field?*
3. *If* $y(x)$ *is the numerical solution of the ode with initial condition* $y(0) = -1$ *then compute* $y(-1)$, $y(1)$, $y(1.5)$, $y(2)$ *and compare your answer with that seen in the graph.*

Exercise 3.2 *Consider the ODE* $y'' = xy + (y')^2$.

1. *In the same plot, plot the solutions of the ODE corresponding to the initial conditions (a)* $y(0) = 1, y'(0) = -3$*; (b)* $y(0) = 2, y'(0) = -3$*; and (c)* $y(0) = 3, y'(0) = -3$*, over the interval* $0 \le x \le 2$.

2. *If* $y(x)$ *is the numerical solution of the above ODE with the initial condition* $y(0) = 3, y'(0) = -3$ *then determine* $y(1)$, $y(1.5)$, $y(3)$. *What happens at* $y(3)$*? Is there an explanation for it?*

Exercise 3.3 *Consider the ODE* $u'' + 3u' + 2u = 3\sin t + e^{-t}$.

1. *Find all solutions of the ODE and verify that they are indeed its solutions.*

2. *Find the solution of the above ODE with the initial condition,* $u(0) = 1,\ u'(0) = 3$. *Then plot this solution over the interval* $0 \le t \le 20$*, first using the* `plot` *command and then using the* `DEplot` *command.*

Exercise 3.4 *For the nonlinear ODE* $x'' - (1 - x^2)x' + x = 0$ *(van der Pol oscillator -* $x \equiv x(t)$*)*

1. *Determine* $x(3)$*, where* $x(t)$ *solves the above ODE with IC* $x(0) = 1,\ x'(0) = 2$.

2. *In the same plot, plot the solutions of the ODE, corresponding to the initial conditions,(i)* $x(0) = 1,\ x'(0) = 1$*, (ii)* $x(0) = 2,\ x'(0) = 3$*, (iii)* $x(0) = 4,\ x'(0) = 2$*, over the range* $0 \le t \le 20$*; choose a small stepsize to get the graphs!*

3.2 Series Solution of Linear 2nd Order ODEs

To find solutions of second order ODEs using power series expansions around $x = 0$ one still uses `dsolve` but with the `'series'` option. For example, to obtain the series solution of $y'' + xy' + (1 - x^2)y = 0$ with the initial conditions $y(0) = 2,\ y'(0) = 1$, up to terms x^7, we use

```
myode := diff( y(x), x$2 ) + x*diff(y(x),x) + (1-x^2)*y(x) = 0;
Order := 8;          series solution will go up to x^7
dsolve( {myode, y(0)=2, D(y)(0)=1}, y(x), 'series' );
```

To find the solutions of the ODE $x^2y'' + xy' + (x^2 - 4)y = 0$, for which $x = 0$ is a regular singular point, one does not prescribe initial conditions (at least at $x = 0$); one is more interested in the form of the general solution near $x = 0$. To find the general series solution of this ODE, up to terms x^5, use

```
myode := x^2 * diff( y(x), x$2 ) + x*diff(y(x),x) + (x^2 - 4)*y(x) = 0;
Order := 6;
mysol := dsolve( myode, y(x), 'series' );
```

Using `mysol` in other Maple calculations is inconvenient because of the `O`(x^6) term; to get rid of this expression and convert the answer to a polynomial, we use

```
convert( mysol, polynom );
```

Exercise 3.5 *Find the power series solution of* $y'' + 3xy' + sin(x)y = x^3$ *up to terms* x^5*. What is the solution when* $y(0) = 1,\ y'(0) = 3$.

3.3 First Order Systems

The commands `dsolve`, `DEplot`, and `dfieldplot` also apply to First Order Systems of ODEs.

3.3.1 Exact Solutions

To find the exact solutions of the system of ODEs

$$\begin{aligned} x' &= x \\ y' &= x + y \\ z' &= x + y + z \end{aligned}$$

with and without initial conditions use

```
mysys :=  diff( x(t),t ) = x(t),                  Shift Return
          diff( y(t),t ) = x(t) + y(t),           Shift Return
          diff( z(t),t ) = x(t) + y(t) + z(t) ;    Return
dsolve( {mysys} , {x(t), y(t), z(t)} ) ;          all solutions - no IC
dsolve( {mysys, x(0)=1, y(0)=2, z(0)=3} , {x(t), y(t), z(t)} ) ;    with IC
```

The last command generated the solution with the initial condition $x(0) = 1, y(0) = 2, z(0) = 3$. Note one important difference between the syntax for solving a system of ODEs and solving a single ODE - *the system of ODEs must be enclosed in parentheses when using any of the ODE commands*, as done for `mysys` above.

3.3.2 Numerical Solutions

To find the numerical solution of the first order system

$$\frac{dx}{dt} = y, \qquad \frac{dy}{dt} = (1 - x^2)y - x$$

corresponding to the initial condition $x(0) = 1,\ y(0) = 1$ use

```
mysys := diff( x(t),t ) = y(t),                      Shift Return
         diff( y(t),t ) = (1-x(t)^2)*y(t) - x(t) ;
mysoln := dsolve( {mysys, x(0)=1, y(0)=1}, {x(t),y(t)}, 'numeric' );
```

As before, use of the `'numeric'` results in a procedure, so `mysoln` is a procedure. To find the value of the solution when $t = 1$ and when $t = 2.5$, we use

```
mysoln(1); mysoln(2.5);
```

3.3.3 Plotting Solutions

To plot the solutions of a first order system of two equations in two unknowns such as the system `mysys` used above, with the ICs (i) $x(0) = 1$, $y(0) = 1$, and (ii) $x(0) = 2$, $y(0) = -1$, as t ranges from 0 to 1, use

```
with(DEtools):
DEplot( {mysys}, {x(t),y(t)}, t=0..1, [ [x(0)=1,y(0)=1], [x(0)=2,y(0)=-1] ] );
```

Note that the picture is drawn in the x, y plane - no t axis. The solutions were traced only for the time interval $0 \le t \le 1$; increase the length of the time interval - what happens to the two solutions? Increase some more. You may also use the options `arrows=..`, `linecolor=..`, `stepsize=..` etc. as done earlier.

In some cases, one is interested in examining how x changes with t without complicating the picture by examining the changes in y. One may do this using the `scene` option.

```
DEplot( {mysys}, {x(t),y(t)}, t=0..1, [ [x(0)=1,y(0)=1], [x(0)=2,y(0)=-1] ],
   scene=[t,x], linecolor=[blue, green]);
DEplot( {mysys}, {x(t),y(t)}, t=0..1, [ [x(0)=1,y(0)=1], [x(0)=2,y(0)=-1] ],
   scene=[t,y], linecolor=[blue,green] );
```

Use ?DEtools to find other options available.

Remark If you are getting confused about where one should use [] parentheses and where to use {} parentheses, I follow the following rule as much as possible and it works correctly most of the time (one can always use Maple's help facility) - *if the order of elements in a list is important then enclose the list in* [ ] *otherwise enclose the list in* { }.

3.3.4 Direction Fields

If we have a first order system of two equations in two unknowns which is *autonomous* (t does not appear explicitly in the equation), such as

$$\frac{dx}{dt} = y, \qquad \frac{dy}{dt} = (1 - x^2)y - x$$

then we can plot its direction field (at each point (x, y) an arrow parallel to the vector $y\tilde{\mathbf{i}} + ((1 - x^2)y - x)\tilde{\mathbf{j}}$). To plot the direction field for the above system over the rectangle $-3 \leq x \leq 3$, $-3 \leq y \leq 3$, use

```
mysys := diff( x(t),t ) = y(t),                          Shift Return
         diff( y(t),t ) = (1-x(t)^2)*y(t) - x(t) ;             Return
dfieldplot( {mysys}, {x(t),y(t)}, t=0..10, x=-3..3, y=-3..3, arrows='SLIM' );
```

Exercise 3.6 *Consider the system*

$$\frac{du}{dt} = v, \qquad \frac{dv}{dt} = u - u^3 - 0.2v$$

1. *Draw the direction field for this system over the range* $-1.5 \leq u \leq 1.5$, $-1 \leq v \leq 1$.
2. *Plot the two solutions of the system corresponding to the ICs* $u(0) = -1.2,\ v(0) = 0.5$, *and* $u(0) = -1.2,\ v(0) = 1$. *Take a long enough time interval and a small enough stepsize so that you get a smooth and informative picture. Do the solutions follow the direction fields?*
3. *Plot the solution of the above system with the IC* $u(0) = -1.2,\ v(0) = 0.5$ in the t, u plane *with a small enough stepsize. Then plot the solution of the ode* $u'' + 0.2u' + u^3 - u = 0$ *with the IC* $u(0) = -1.2,\ u'(0) = 0.5$. *Compare the two plots - do you see something unusual? Can you explain it?*

3.4 Laplace and Inverse Laplace Transforms

The functions $u_c(t)$ and $\delta(t - c)$ are entered in Maple as $Heaviside(t - c)$ and $Dirac(t - c)$. Below we show how to use them, plot them, and take their Laplace transforms. These commands are in the inttrans (integral transforms) package.

To plot the function

$$f(t) = \begin{cases} 3t+1, & \text{if } 1 \leq t \leq 5 \\ 1 & \text{otherwise} \end{cases}$$

```
with(intttrans);
f := 1 + 3*t*( Heaviside(t-1) - Heaviside(t-5) );
plot( f, t=0..10, discont=true );
```

Note the use of the `discont=true` option in `plot` because the graph is discontinuous.

To compute the Laplace transform of $t + 4\delta(t-1)$ we use

```
g := t + 4*Dirac(t-1);
G := laplace( f+g, t, s);              K is laplace transform of f+g
```

To find the inverse Laplace transform

```
F := (2*s - 5) + 2/( (s-1)^2 + 4 );
G := s^3 - 3*s^2 + 4;
invlaplace( F/G, s, t);  inverse laplace transform of F/G
```

Note the position of the s and t in the *laplace* and *invlaplace* commands - beginners usually make errors there.

Maple also has a command to compute the partial fraction decomposition. To compute the partial fraction decomposition of F/G where F and G are as above, we use

```
convert( F/G, parfrac, s );      partial fraction decomposition of f/g
```